
25

PROGRAMAREA APLICAŢIILOR WINDOWS ÎN LIMBAJ DE ASAMBLARE

25.1 Interfaţa de programare a aplicaţiilor Windows

API (Application Programming Interface) pentru sistemele Win32 este implementat în biblioteci cu legare dinamică.. Aceste biblioteci sunt separate de programul executabil. Spre deosebire de DOS, API-urile nu sunt accesate prin întreruperi. 

Codul funcţiilor Windows este stocat în fişiere din afara programului, numite DLL (Dynamic Link Libraries) - biblioteci cu legături dinamice. Atunci când este rulat un program Windows, interconectarea acestuia cu sistemul de operare se face prin procesul de legare dinamică. Un fişier executabil al unui program Windows conţine referinţe la biblioteci cu legături dinamice. Atunci când un program pentru Windows este încărcat în memorie, apelurile de funcţii sunt rezolvate astfel încât să indice intrările în funcţiile din DLL-urile utilizate, care sunt încărcate în memorie, în cazul în care nu se află deja acolo.

Principalele biblioteci cu legare dinamică sunt kernel32.dll, user32.dll şi gdi32.dll. Biblioteca cu legare dinamică kernel32.dll conţine funcţiile API pentru gestionarea memoriei şi a proceselor. Biblioteca cu legare dinamică user32.dll controlează aspectele interfeţei cu utilizatorul ale programului. Biblioteca cu legare dinamică gdi32.dll este responsabilă cu operaţiile grafice. În afară de aceste biblioteci principale se pot utiliza în program şi alte biblioteci cu legare dinamică.

Există două categorii de funcţii API: unele pentru ANSI  şi altele pentru Unicode. Şirurile ANSI sunt vectori de caractere terminate cu NULL. Un caracter ANSI are dimensiunea de 1 byte. Un caracter Unicode are dimensiunea de 2 octeţi. Numele funcţiilor API pentru ANSI se termină cu caracterul "A" (de exemplu MessageBoxA), iar numele funcţiilor API pentru Unicode se termină cu W (MessageBoxW). În mod nativ Windows 95 suportă ANSI iar Windows NT Unicode. În general în fişierele cu prototipurile funcţiilor există declaraţii echivalente ale acestora în care se elimină caracterul A, respectiv W (de exemplu MessageBox).
25.2 Organizarea memoriei în Windows 9x

Programele Win32 rulează în modul protejat, acest mod existând începând cu microprocesorul 80286. Windows rulează fiecare program Win32 în spaţii virtuale separate. Aceasta înseamnă că fiecare program Win32 va avea propriul spaţiu de adresare de 4 GB. Deoarece programele Win32 au 4 GB de memorie adresabilă, pointerii pot adresa zone de memorie cuprinse între 00000000H şi FFFFFFFFH. În Windows 9x , fiecare proces obţine propriul spaţiu de memorie de 4 GB, spaţiu care este în totalitate al său. De fapt el partajează împreună cu alte aplicaţii care rulează, primii 2 GB de memorie. Este important să se considere că cei 4GB ai spaţiului procesul sunt numai şi numai ai acestuia. Acest model de memorie se numeşte flat.

Secţiunea cea mai joasă a memoriei (regiunea de 64 KB) este rezervată în Windows 95 pentru atribuirile de pointeri nuli. Restul memoriei, până la 4 MB este rezervată de sistem pentru compatibilitatea cu aplicaţiile Win16 şi cu aplicaţiile MS-DOS.

Programele Win32 sunt încărcate la adresa 00400000H (adresa de bază a aplicaţiilor Win32 în memorie), deci spaţiul de lucru al programului este până la 7FFFFFFFH. Începând cu adresa 80000000H, Windows 9x încarcă propriile DLL-uri de sistem. Începând cu adresa C0000000H până la sfârşitul celor 4 GB se încarcă sistemul de operare Windows 9x precum şi programele pentru dispozitive virtuale.

Puţine calculatoare au 4GB de memorie. Datorită acestui lucru, proiectanţii de hardware au introdus ideea de memorie virtuală şi memorie fizică. Memoria virtuală corespunde celor 4 GB de memorie adresabilă, memoria fizică este limitată de memoria RAM disponibilă în calculator şi de spaţiul pe hard disk. Memoria este împărţită în pagini, dimensiunea paginii fiind optimizată în funcţie de suportul hardware, pe sistemele 80x86, mărimea unei pagini fiind de 4 KB. 


Figura 25.1 – Organizarea memoriei în Windows 9x

Sistemul de gestiune a memoriei determină paginile localizate în memoria virtuală şi în RAM şi furnizează un mecanism de conversie a adreselor din memoria virtuală în adrese fizice. Programul de gestiune a memoriei poate muta sau comuta paginile de memorie dacă este nevoie de spaţiu suplimentar de memorie fizică. Windows schimbă memorie în şi afară din fişierul de paginare de pe hard disk. Dimensiunea fişierului de paginare împreună cu dimensiunea memoriei RAM este considerată memoria fizică.

Când este rulat un program, fişierul EXE de pe disc (numit imaginea programului) este considerat ca făcând parte din fişierul de paginare de pe hard disk. În acest fel se economiseşte timpul necesar încărcării fişierului executabil în memoria RAM. Imaginea fişierului EXE de pe disc este considerată parte a fişierului de paginare şi secţiunile de cod (nu şi de date) sunt citite de pe disc atunci când este nevoie de ele.

25.3 Încărcarea programelor Win32

Aplicaţiile Win32 rulează în segmente cu adresare pe 32 de biţi, utilizând modelul de memorie FLAT. Astfel, programul este în mod automat în modul protejat. Adresele generate şi utilizate de program se numesc adrese liniare. Registrele de segment CS, DS, ES şi SS sunt iniţializate astfel că nu mai contează ce segment este utilizat pentru a adresa o locaţie dată (o adresă liniară). Adresa de bază la care este încărcat programul în memorie este 0040000h.

La lansarea în execuţie a programelor, valorile iniţiale ale regiştrilor sunt:

· CS:EIP – începutul programului

· SS:ESP – începutul stivei

· DS=ES=SS – selectorul de date

· FS=TIB (Thread Information Block)

· GS=0, selectorul nul

· CS şi DS sunt mapate pe aceeaşi adresă liniară

· Indicatorul de direcţie, DF, este şters.

Sistemul Windows utilizează intern regiştrii esi, edi, ebp şi ebx şi de aceea presupune că valorile acestor regiştri nu se vor schimba. Se recomandă deci în cazul utilizării acestor regiştri în cadrul unor funcţii callback să se salveze conţinutul lor înainte de modificare şi să se restaureze înainte de a da controlul sistemului Windows. O funcţie callback este o funcţie scrisă de programator, funcţie ce este apelată de Windows. Un exemplu de astfel de funcţie este funcţia fereastră.

25.4 Structura unui program în limbaj de asamblare


Un program Win32 scris în limbaj de asamblare are următoarea structură:

.386

.model flat, stdcall

option casemap :none  

include \masm32\include\windows.inc

includelib \masm32\lib\user32.lib

.data

;

;datele iniţializate

;

.data?

;

;datele neiniţializate

;

.const

;

;constantele

;

.code

eticheta_de_inceput

;

;codul propriu zis

;

end eticheta_de_inceput
Directiva .386 anunţă asamblorul că se va utiliza setul de instrucţiuni specifice procesorului 80386. Se pot utiliza şi directivele .486, .586. Pentru fiecare model al microprocesorul sunt două forme aproape identice ale acestei directive: .386/.386p, .486/.486p. Versiunile cu p sunt necesare atunci când programul utilizează instrucţiuni privilegiate. Instrucţiunile privilegiate sunt rezervate de microprocesor/sistem de operare în modul protejat. Ele pot fi utilizate numai de cod protejat, cum ar fi programele pentru echipamente virtuale.


Directiva .model specifică modelul de memorie al programului. Sub Win32 există numai modelul de memorie flat. 


În directiva .model mai este precizata şi convenţia de transmitere a parametrilor (stdcall). Convenţia de transmitere a parametrilor specifică ordinea de transmitere a parametrilor (de la stânga la dreapta sau de la dreapta la stânga) şi de asemenea cine va echilibra stiva după apelul funcţiei.

Sub Win16 există două tipuri de convenţii de apel: C şi PASCAL. 

Convenţia de apel C transmite parametri de la dreapta la stânga, cel mai din dreapta parametru fiind pus primul pe stivă. Apelatorul este responsabil cu echilibrarea stivei după apel. De exemplu apelul unei funcţii f(int a, int b) utilizând convenţia C de transmitere a parametrilor în limbaj de asamblare va fi:

push [b]

push [a]

call f

add sp, 8 ; apelatorul echilibrează stiva
Convenţia de apel PASCAL presupune transmiterea parametrilor de la stânga la dreapta, funcţia apelată fiind responsabilă cu echilibrarea stivei după apel. 

Convenţia de apel C este utilă atunci când nu se ştie exact numărul de parametri ce vor fi transmişi funcţiei. O astfel de funcţie este wsprinf. Nu se ştie dinainte câţi parametri vor fi puşi pe stivă, deci în cadrul funcţiei nu se poate realiza echilibrarea stivei.

STDCALL este un hibrid între convenţiile de transmitere a parametrilor C şi PASCAL. Parametri sunt puşi pe stivă de la dreapta la stânga (ca în convenţia C), însă funcţia apelata este responsabilă cu echilibrarea stivei (convenţia PASCAL). Platforma Win32 utilizează exclusiv convenţia STDCALL cu excepţia funcţiei wsprintf.

Directiva option casemap :none  anunţă MASM că etichetele sunt case-sensitive.

Directiva include \masm32\include\windows.inc inserează fişierul windows.inc în acest punct în programul sursă. Acest fişier conţine declaraţiile pentru majoritatea constantelor şi structurilor utilizate de API. După această directivă se urmează includerea fişierelor ce conţin antetele funcţiilor utilizate din bibliotecile respective. De exemplu dacă se utilizează funcţia MessageBox se va scrie include \masm32\include\user32.inc, având în vedere că această funcţie există în user32.dll.

Directiva includelib \masm32\lib\user32.lib informează asamblorul ce biblioteci de import sunt utilizate. Când asamblorul întâlneşte această directivă, el pune o comandă a editorului de legături în fişierul obiect astfel încât editorul de legături să ştie ce biblioteci utilizează programul şi de unde să le ia.

Secţiunea .data conţine datele iniţializate ale programului. Secţiunea .data? conţine datele neiniţializate ale programului. Aceasta are avantajul că fişierul executabil nu îşi măreşte dimensiunea corespunzător cu mărimea datelor neiniţializate. Ea doar informează asamblorul cât spaţiu de memorie este necesar când programul este încărcat în memorie. Secţiunea .const conţine declaraţiile de constante din program. Aceste nu pot fi modificate în program. 

Secţiunea .code conţine codul programului. Eticheta_de_inceput este o etichetă ce marchează începutul codului programului. Prima instrucţiune executabilă se află după eticheta_de_inceput. Codul trebuie să fie între eticheta_de_inceput şi end eticheta_de_inceput.

Pe lângă programul sursă, dacă în program se utilizează resurse ca meniuri, pictograme, căsuţe de dialog, acestea se definesc într-un fişier special, numit fişier de resurse, fişier ce are extensia .rc. Acesta este un fişier text în care se descriu resursele şi se face asocierea cu fişiere existente pe disc. De exemplu, pentru resursa pictogramă este linia:

ID_ICON ICON "WHAT1.ICO"
25.5 Programarea sub Windows

Majoritatea programelor Windows au o fereastră principală asociată. Atunci când programul este lansat în execuţie, Windows creează o coadă de mesaje pentru programul respectiv. În această coadă de mesaje sunt păstrate mesajele trimise către toate ferestrele pe care le creează programul. Programul conţine o mică secvenţă de cod numită bucla de mesaje care preia mesajele din coada de mesaje şi le distribuie funcţiilor fereastră corespunzătoare. Anumite mesaje sunt trimise direct funcţiei fereastră fără a mai fi puse în coada de mesaje.

Paşii necesari creării unei ferestre sunt:

· obţinerea identificatorului instanţei programului

· obţinerea liniei de comandă (opţional)

· înregistrarea clasei fereastră (window class)

· crearea ferestrei

· afişarea ferestrei

· actualizarea zonei client a ferestrei

· intrarea în bucla de mesaje infinită

· dacă sunt mesaje pentru fereastră ele sunt prelucrate de o funcţie specializată numită şi funcţia fereastră

· terminarea programului

Obţinerea identificatorului instanţei programului se realizează prin apelul funcţiei GetModuleHandle. Funcţia returnează identificatorul instanţei curente a programului. Prototipul funcţiei este:

HMODULE GetModuleHandle(LPCTSTR lpModuleName);

Prin transmiterea ca parametru a valorii NULL se obţine identificatorul instanţei curente. După apelul funcţiilor API, rezultatul returnat se găseşte în registrul EAX.


.data?

;...

hinst
dd
?


;...


.code

;...

push 0 

call GetModuleHandle

mov hinst,eax

;...

Obţinerea liniei de comandă se face cu funcţia GetCommandLine. Obţinerea liniei de comanda nu este necesară decât dacă programul prelucrează parametrii din linia de comandă.

Înregistrarea clasei fereastră se realizează având în vedere faptul că o fereastră este întotdeauna creată pe baza unei clase de fereastră. Aceasta identifică procedura de fereastră care prelucrează toate mesajele transmise către fereastră. Pe baza aceleiaşi clase se pot crea mai multe ferestre.

Înainte de crearea ferestrei programului trebuie înregistrată clasa fereastră prin apelul funcţiei RegisterClassEx ce are un singur parametru: un pointer la o structură de tipul WNDCLASSEX.

Structura WNDCLASSEX are următorii membri:

WNDCLASSEX struc

cbSize 

dd ?
; dimensiunea structurii WNDCLASSEX

style  

dd ?
; stilul clasei fereastră

lpfnWndProc  
dd ?
; pointerul la funcţia fereastră

cbClsExtra  
dd ?
; informaţii suplimentare





;pentru clasă

cbWndExtra  
dd ?
; informaţii suplimentare





;pentru fereastră

hInstance  
dd ?
; identificatorul instanţei





;curente

hIcon  

dd ?
; identificatorul pictogramei





;asociata clasei

hCursor  

dd ?
; cursorul asociat clasei

hbrBackground  
dd ?
; culoarea de fundal a





;ferestrelor din clasă

lpszMenuName  
dd ?
; pointer la şirul cu denumirea





;meniului

lpszClassName  
dd ?
; pointer la şirul cu numele





;clasei

hIconSm  

dd ?
; identificatorul pictogramei





;aplicatiei

WNDCLASSEX ends
Al treilea şi penultimul câmp (lpfnWndProc şi lpszClassName) sunt cele mai importante, ele conţinând adresa procedurii fereastră folosită de toate ferestrele create pe baza acestei clase, respectiv numele clasei fereastră. Celelalte câmpuri  descriu caracteristicile tuturor ferestrelor create pe baza acestei clase.

Înainte de apelul funcţiei RegisterClassEx trebuie iniţializate câmpurile variabilei de tip WNDCLASSEX. Iniţializarea câmpurilor se realizează în următoarea secvenţă de cod:

;...

.data?

;...

wcex WNDCLASSEX <?>;var de tip structura clasa fereastra

NumeFer     db  'Programare sub Windows',0 ;Numele ferestrei

hinst  dd  ? ;identificatorul instantei curente

;...

.code

;...

     
;completarea cimpurilor structurii fereastra

     
mov wcex.cbSize, size WNDCLASSEX

     
mov wcex.style, CS_VREDRAW or CS_HREDRAW or CS_DBLCLKS

mov wcex.lpfnWndProc, offset WndProc

mov wcex.cbClsExtra,0 

mov wcex.cbWndExtra,0 

push hinst

pop wcex.hInstance

;se incarca pictogtama, definita in fisierul .rc

push ID_ICON ;icon id

push hinst

call LoadIcon

mov wcex.hIcon,eax 

;se incarca cursorul sageata standard

push IDC_ARROW

push NULL

call LoadCursor

mov wcex.hCursor,eax

mov wcex.hbrBackground,COLOR_WINDOW+1 

mov wcex.lpszMenuName,0

mov wcex.lpszClassName,offset NumeFer 

mov wcex.hIconSm,0 

Se observă că pentru iniţializarea câmpurilor hIcon şi hCursor ale structurii WNDCLASSEX s-au apelat funcţiile LoadIcon şi LoadCursor. Funcţia LoadIcon are prototipul:

HICON LoadIcon( HINSTANCE hInstance, // identificatorul instantei aplicatiei

LPCTSTR lpIconName // numele identificatorul resursei pictograma 

);
Funcţia returnează un identificator al pictogramei asociat resursei pictogramă definită în fişierul de resurse. În program se transmite ID_ICON, definit în fişierul de resurse.

Funcţia LoadCursor:

HCURSOR LoadCursor( HINSTANCE hInstance, // identificatorul instantei aplicatiei 

LPCTSTR lpCursorName // numele/identificatorul resursei cursor

); 
returnează identificatorul de cursor asociat cursorului definit în fişierul de resurse, sau unuia din cursoarele standard. În aplicaţie s-a utilizat cursorul standard de tip săgeată, identificat prin IDC_ARROW. Atunci când se utilizează cursoare standard, parametrul hInstance al funcţiei LoadCursor va fi 0.

Apelul funcţiei RegisterClassEx prin care se înregistrează clasa fereastră:

.code

;...

;Inregistrarea clasei fereastra


push offset wcex

call RegisterClassEx

;...
În cazul în care nu s-a reuşit înregistrarea clasei de fereastră, funcţia RegisterClassEx returnează valoarea 0.

Sunt definite câteva clase standard de ferestre înregistrate: "BUTTON", "EDIT", "LISTBOX"  şi altele, clase fereastră ce nu mai trebuie înregistrate, ele având predefinite funcţiile fereastră şi tratarea anumitor mesaje. Pe baza acestor clase de ferestre standard se pot crea ferestre de acel tip, modificându-se anumite caracteristici, modul de funcţionare rămânând acelaşi.

Clasa fereastră defineşte caracteristicile generale ale unei ferestre, permiţând folosirea acesteia pentru crearea mai multor ferestre. Când se creează o fereastră cu funcţia CreateWindowEx se pot specifica mai multe detalii despre fereastra respectivă (cum ar fi titlul ferestrei, stilul acesteia, poziţia, meniul asociat şi dimensiunea iniţială).

Prototipul funcţiei CreateWindowEx este:

HWND CreateWindowEx(

DWORD dwExStyle,
   // stilul extins al ferestrei

LPCTSTR lpClassName, // pointer către numele clasei înregistrate

LPCTSTR lpWindowName,// pointer către numele ferestrei

    
DWORD dwStyle,
   // stilul ferestrei

    
int x,

   // poziţia orizontală a ferestrei

    
int y,

   // poziţia verticală a ferestrei

    
int nWidth,

   // lăţimea ferestrei

    
int nHeight,
   // înălţimea ferestrei

    
HWND hWndParent,
   // indicator către fereastra părinte

    
HMENU hMenu,
   // indicator către meniu sau fereastra copil

    
HINSTANCE hInstance, // indicatorul instanţei curente a programului

    
LPVOID lpParam 
   // pointer către datele de creare a ferestrei

   );

Funcţia returnează identificatorul ferestrei create.

Apelul funcţiei CreateWindowEx este:


.data?

;...

hwnd    dd  ?
;identificatorul de fereastra

hinst   dd  ?
;identificatorul instantei curente 

hmeniu  dd  ?
;identificatorul meniului

;...


.const

;...

NumeFer     db  'Programare sub Windows',0 ;Numele ferestrei

NumeApp     db  'Programare in Win32 ASM',0;Numele aplicatiei

;...

.code

;...

    ;Crearea ferestrei

     push ID_MENIU

     push hinst

     call LoadMenu  

     mov hmeniu,eax

     push 0                        ; lpParam 

     push hinst                    ; hInstance 

     push hmeniu                   ; identificator meniu 

     push 0                        ; identificator parinte 

     push CW_USEDEFAULT            ; inaltime 

     push CW_USEDEFAULT            ; latime 

     push CW_USEDEFAULT            ; y 

     push CW_USEDEFAULT            ; x 

     push WS_OVERLAPPEDWINDOW      ; stilul

     push offset NumeApp           ; titlul ferestrei

     push offset NumeFer           ; numele clasei 

     push  WS_EX_OVERLAPPEDWINDOW  ; stilul extins 

     call CreateWindowEx

     mov hwnd,eax
;...
Pentru a  asocia resursa meniu definită în fişierul de resurse se apelează funcţia LoadMenu ce are ca parametri identificatorul instanţei curente a aplicaţiei şi numele meniului definit în fişierul de resurse. Funcţia returnează identificatorul de meniu asociat.

După apelul funcţiei CreateWindowEx, fereastra a fost creată însă nu a fost şi afişată pe ecran. Pentru afişarea ferestrei sunt necesare încă două apeluri de funcţii: ShowWindow respectiv UpdateWindow.

Funcţia ShowWindow are prototipul:

BOOL ShowWindow(HWND hWnd, // identificatorul ferestrei

   int nCmdShow // modul de afişare a ferestrei); 
Apelul funcţiei în cadrul programului:

push SW_SHOWNORMAL

     
push hwnd

     
call ShowWindow
Funcţia UpdateWindow redesenează zona client a ferestrei. Prototipul acesteia este:

BOOL UpdateWindow(HWND hWnd // identificatorul ferestrei 

); 
În program, apelul se realizează astfel:

     
push hwnd

     
call UpdateWindow
După apelul funcţiei UpdateWindow, fereastra devine vizibilă pe ecran.

Programul trebuie să poată prelucra intrările de la tastatură şi de mouse. Windows creează o coadă de mesaje asociată fiecărui program. Când apare un eveniment, acesta este transformat de Windows într-un mesaj care este pus în coada de mesaje a programului. Mesajele sunt preluate din coada de mesaje prin intermediul buclei de mesaje. În pseudocod bucla de mesaje se reprezintă astfel:

Aşteaptă_mesaj:


Dacă este apare un mesaj, preia mesajul


Dacă mesajul este QUIT Atunci Ieşire


Apelează TranslateMessage


Apelează DispatchMessage


Trimite mesajul către procedura fereastră


Înapoi la Aşteaptă_mesaj

Ieşire
Funcţia GetMessage preia un mesaj din coada de mesaje. Prototipul funcţiei GetMessage:

BOOL GetMessage(

    LPMSG lpMsg,

// adresa structurii  cu mesajul

    HWND hWnd,

// identificatorul ferestrei

    UINT wMsgFilterMin,
// primul mesaj

    UINT wMsgFilterMax 
// al doilea mesaj

   );
Primul parametru este un pointer la o structură de tip MSG. Windows completează câmpurile structurii de mesaje cu următorul mesaj din coada de aşteptare. Structura este MSG este:

MSG STRUC

  hwnd      DWORD ? ; identificatorul ferestrei căreia îi    este destinat mesajul

  message   DWORD ? ; identificatorul mesajului

  wParam    DWORD ? ; parametru dependent de mesaj

  lParam    DWORD ? ; parametru dependent de mesaj

  time      DWORD ? ; momentul inserării mesajului în coada     de mesaje

  pt        POINT <>; poziţia mouse-ului în momentul inserării mesajului

MSG ENDS
Dacă în câmpul message al structurii este transmisă o valoare diferită de WM_QUIT, funcţia GetMessage returnează o valoare diferită de zero. Mesajul WM_QUIT determină ieşirea din bucla de mesaje.

În cadrul buclei de mesaje se apelează două funcţii. TranslateMessage transmite structura MSG sistemului de operare pentru convertirea unor mesaje de la tastatură. Prototipul acesteia este:

BOOL TranslateMessage( CONST MSG *lpMsg // pointer la o structură MSG 

); 
Funcţia DispatchMessage, al cărei prototip este:

LONG DispatchMessage(

CONST MSG *lpmsg // pointer către structura ce conţine mesajul

   );

retransmite structura MSG sistemului de operare.

Windows trimite apoi mesajul către funcţia fereastră corespunzătoare în vederea prelucrării acestuia.

Bucla de mesaje este implementată astfel:

;Bucla de mesaje

start_bucla: 

     push 0 

; wMsgFilterMax 

     push 0 

; wMsgFilterMin 

     push NULL

; 0 - toate ferestrele 

     push offset msg 
; lpMsg 

     call GetMessage 
; returneaza FALSE pentru WM_QUIT 

     or eax,eax 

     jz iesire 

     push offset msg 

     call TranslateMessage

     push offset msg 

     call DispatchMessage 

     jmp start_bucla

Terminarea programului se face prin apelul funcţiei ExitProcess. Prototipul acesteia este:

VOID ExitProcess(UINT uExitCode 
// codul de ieşire al procesului 

 );
Secvenţa de cod în care se face apelul funcţiei este:

push  0 
;codul de iesire returnat de aplicatie 

call ExitProcess

Funcţia fereastră este responsabilă de afişarea în zona client a ferestrei, precum şi de tratarea mesajelor provenite de la tastatură şi mouse. Funcţia fereastră poate avea orice nume, iar un program pentru Windows poate avea mai multe funcţii fereastră, asociate ferestrelor definte în cadrul acestuia. Funcţia fereastră este întotdeauna asociată unei clase fereastră.
Funcţia fereastră are următorul prototip:

LRESULT WndProc(HWND hwnd, //identificatorul ferestrei

    UINT msg,  //identificatorul mesajului

    WPARAM wparam, //parametru asociat mesajului

    LPARAM lparam  //parametru asociat mesajului

   

    );

Parametrii funcţiei corespund primelor patru câmpuri ai structurii MSG. Fiecare mesaj recepţionat  de fereastră este identificat printr-un număr (parametrul msg al funcţiei fereastră). În fişierele antet din Windows acestora le corespund identificatori prefixaţi de WM_ (de la window message)

Pentru tratarea diverselor mesaje se utilizează o instrucţiune switch pentru determinarea mesajelor primite de fereastră şi a modului de prelucrare a acestuia. Atunci când prelucrează un mesaj funcţia fereastră trebuie să returneze valoarea zero. Toate mesajele ce nu sunt prelucrate de funcţia fereastră sunt transmise funcţie DefWindowProc.  Prototipul funcţiei DefWindowProc este:

LRESULT DefWindowProc(

    HWND hWnd,

// identificatorul ferestrei

    UINT Msg,

// mesajul

    WPARAM wParam,
// primul parametru al mesajului

    LPARAM lParam 
// al doilea parametru al mesajului

   );
Un mesaj ce trebuie tratat este WM_DESTROY care este trimis atunci când utilizatorul  doreşte să închidă fereastra. Mesajul este tratat standard, prin apelul funcţiei PostQuitMessage, ce are prototipul:

VOID PostQuitMessage(

    int nExitCode 
// codul de ieşire

   );
Implementarea funcţiei fereastră, care tratează mesajul WM_DESTROY este:

WndProc proc 

     ;parametrii de pe stiva sunt:

     ;esp+4  --> hwnd

     ;esp+8  --> msg

     ;esp+12 --> wParam

     ;esp+16 --> lParam

cmp dword ptr [esp+8],WM_DESTROY

     je msg_wm_destroy 

     jmp DefWindowProc

msg_wm_destroy:

     push  0

     call  PostQuitMessage

     xor   eax,eax

     ret 16

WndProc endp

Celelalte mesaje sunt tratate de funcţia standard.

În programul scris în limbaj de asamblare nu are importanţă dacă numele funcţiei fereastră este WndProc, acest lucru fiind la latitudinea programatorului.

25.6 Asamblarea şi editarea de legături

Pentru realizarea programului s-a utilizat MASM32. Acesta se găseşte la adresa http://www.pbq.com.au/home/hutch/masm.htm.

Asamblarea programului sursă se realizează astfel:

ml /c /coff /Cp program.asm 

Dacă se folosesc fişiere de resurse, acestea trebuiesc compilate cu un compilator de resurse (de exemplu rc.exe), obţinându-se un fişier de resurse obiect, fişier ce va fi link-editat împreună cu fişierul obiect al programului. Editarea de legături se realizează utilizând următoare linie de comandă:

link /SUBSYSTEM:WINDOWS /LIBPATH:c:\masm32\lib program.obj

25.7 Exemplu de program Win32
În continuare se prezintă un program Windows, program care afişează o fereastră, în funcţia fereastră tratându-se mesajele WM_COMMAND (alegerea unei opţiuni din meniu), WM_LBUTTONDOWN (executarea unui clic de mouse în zona client a ferestrei) şi WM_DESTROY (la închiderea ferestrei). 

Fereastra are asociat un meniu având submeniul "Fisiere" cu o un singur articol: "Iesire" şi submeniul "Help" cu articolul "Despre...". Când se selectează "Iesire" se va închide fereastra, iar când se selectează "Despre..." se afişează o căsuţă de dialog.

.386

.model flat, stdcall

option casemap :none  

include \masm32\include\windows.inc

include \masm32\include\kernel32.inc

include \masm32\include\user32.inc

include res.equ

includelib \masm32\lib\user32.lib

includelib \masm32\lib\kernel32.lib

.data

.const

NumeFer     db  'Programare sub Windows',0 
;Numele ferestrei

NumeApp     db  'Programare in Win32 ASM',0 
;Numele aplicatiei

MesajDespre db  'Paul Pocatilu, 1999',0

Mesaj1      db  'Mouse click',0

.data?

wcex WNDCLASSEX <?>
;var de tip structura clasa fereastra

msg
  MSG
<?>

;Var de tip MSG

hwnd    dd  ?

;identificatorul de fereastra

hinst   dd  ?

;identificatorul instantei curente 

hmeniu  dd  ?           ;identificatorul de meniu

.code

start:

     ;obtinerea hInstance

     push 0 

     call GetModuleHandle

     mov hinst,eax

     ;completarea campurilor structurii fereastra

     mov wcex.cbSize, size WNDCLASSEX

     mov wcex.style, CS_VREDRAW or CS_HREDRAW or CS_DBLCLKS

     mov wcex.lpfnWndProc, offset WndProc

     mov wcex.cbClsExtra,0 

     mov wcex.cbWndExtra,0 

     push hinst

     pop wcex.hInstance

     ;se incarca iconul, definit in fisierul .rc

     push ID_ICON ;icon id

     push hinst

     call LoadIcon

     mov wcex.hIcon,eax 

     ;se incarca cursorul sageata standard

     push IDC_ARROW

     push NULL

     call LoadCursor

     mov wcex.hCursor,eax

     mov wcex.hbrBackground,COLOR_WINDOW+1 

     mov wcex.lpszMenuName,0

     mov wcex.lpszClassName,offset NumeFer 

     mov wcex.hIconSm,0 

     ;Inregistrarea clasei fereastra


     push offset wcex

     call RegisterClassEx

     ;s-a inregistrat fereastra?

     or eax,eax

     jz iesire 

     ;Crearea ferestrei

     push ID_MENIU

     push hinst

     call LoadMenu  

     mov hmeniu,eax

     push 0                        ; lpParam 

     push hinst                    ; hInstance 

     push hmeniu                   ; identificator meniu 

     push 0                        ; identificator parinte 

     push CW_USEDEFAULT            ; inaltime 

     push CW_USEDEFAULT            ; latime 

     push CW_USEDEFAULT            ; y 

     push CW_USEDEFAULT            ; x 

     push WS_OVERLAPPEDWINDOW      ; stilul

     push offset NumeApp           ; titlul ferestrei

     push offset NumeFer           ; numele clasei 

     push  WS_EX_OVERLAPPEDWINDOW  ; stilul extins 

     call CreateWindowEx

     mov hwnd,eax

     ;se afiseaza fereastra

     push SW_SHOWNORMAL

     push hwnd

     call ShowWindow

     ;se actualizeaza zona client

     push hwnd

     call UpdateWindow

;Bucla de mesaje

start_bucla: 

     push 0 

; wMsgFilterMax 

     push 0 

; wMsgFilterMin 

     push NULL

; 0 - toate ferestrele 

     push offset msg 
; lpMsg 

     call GetMessage 
; returneaza FALSE pentru WM_QUIT 

     or eax,eax 

     jz iesire 

     push offset msg 

     call TranslateMessage

     push offset msg 

     call DispatchMessage 

     jmp start_bucla

iesire:

     ;Terminarea programului

     push  0 
   ; codul de iesire returnat de aplicatie 

     call ExitProcess

;Functia Fereastra

WndProc proc 

     cmp dword ptr [esp+8],WM_COMMAND

     je msg_wm_command   

     cmp dword ptr [esp+8],WM_DESTROY

     je msg_wm_destroy 

     cmp dword ptr [esp+8],WM_LBUTTONDOWN

     je msg_wm_lbuttondown

     jmp DefWindowProc

msg_wm_command:

     ;se testeaza alegerea optiunilor din meniu

     cmp dword ptr [esp+12],ID_FISIERE_IESIRE

     je msg_wm_destroy

     cmp dword ptr [esp+12],ID_HELP_DESPRE

     je help_despre

     xor eax,eax

     ret 16

msg_wm_lbuttondown:

     ;s-a facut click in zona client a ferestrei

     ;se afiseaza o casuta de dialog

     ;avind ca titlu NumeApp si mesaj Mesaj1

     push MB_OK

     push offset NumeApp

     push offset Mesaj1

     push NULL

     call MessageBox

     xor eax,eax

     ret 16

help_despre:

     ;s-a ales optiunea Despre... din meniul Help

     ;se afiseaza o casuta de dialog

     ;avind ca titlu NumeApp si mesaj MesajDespre

     push MB_OK

     push offset NumeApp

     push offset MesajDespre

     push NULL

     call MessageBox

     xor eax,eax

     ret 16

msg_wm_destroy:

     ;s-a inchis fereastra

     push  0

     call  PostQuitMessage

     xor   eax,eax

     ret 16

WndProc endp

end start
Fişierul de resurse asociat are următorul conţinut:

#define ID_ICON  101

#define ID_MENIU 102

ID_ICON ICON "MAIN.ICO"

ID_MENIU MENUEX

BEGIN

    POPUP "&Fisiere", , , 0

    BEGIN

        MENUITEM "&Iesire", ID_FISIERE_IESIRE

    END

    POPUP "&Help", , , 0

    BEGIN

        MENUITEM "&Despre...", ID_HELP_DESPRE

    END

END
La lansarea în execuţie, programul va afişa o fereastră, dimensiunile şi poziţia acesteia fiind generate de sistemul Windows (prin utilizarea CW_USEDEFAULT). Prin apăsarea butonului stâng al mouse-ului în zona client a ferestrei, prin tratarea evenimentului WM_LBUTTONDOWN se va activa o căsuţă de dialog. Închiderea ferestrei se realizează fie în mod clasic, fie prin alegerea opţiunii Iesire din meniul Fisiere.

